i.r.e.m.
UNIVERSITE PARIS VII

BASIC, RIEMANN, DARBOUX ....

ILLUSTRATION DE L'INTEGRALE SUR UN MICROORDINATEUR

PAR P. JARRAUD

cahier de didactique des mathematiques

numéro 27

0. - Introduction

Le programme qui suit cherche à illustrer certains aspects de la théorie de l'intégrale telle qu'elle est souvent pratiquée en DEUG et à aider les étudiants à mieux appréhender les problèmes d'approximation et de convergence liés à l'existence et au calcul d'une intégrale. Le fonctionnement est automatique (les choix se font par des "menus", il n'y a pas de ligne de programme à modifier) et interactif : l'utilisateur choisit librement : fonction à intégrer, bornes, méthode d'intégration, pas d'intégration, façon de cadrer le graphe, etc...

A priori ce programme est conçu pour, en illustration d'un cours théorique, être utilisé directement par les étudiants au cours d'une séance de travaux dirigés sur micro-ordinateur. Lors de la présentation aux étudiants de la SECTION EXPERIMENTALE de DEUG SSM 1 de l'Université Pierre et Marie Curie (Paris 6) cela n'a pas été le cas, faute de matériel et il a fallu se contenter d'une démonstration faite à tout un groupe de T.D. sur deux téléviseurs, accentuant ainsi le côté "spectacle" au détriment de l'aspect "découverte individuelle".

Je tiens à remercier les collègues de la section expérimentale pour leur aide compréhensive et pour m'avoir fourni l'occasion de présenter ce programme à des étudiants.

1. - Convergence d'une intégrale

Une fonction $x \longmapsto f(x)$ étant donnée sur un intervalle $[a\ b]$, on cherche à définir, si cela est possible, l'intégrale $I(f) = \int_a^b f(x)\,dx$.

Une approche classique en DEUG est de définir d'abord les intégrales de fonctions en escalier puis d'approcher la fonction à intégrer par des fonctions en escalier f_i et de dire que $I(f)$ existe si les $I(f_i)$ ont une limite quand les f_i tendent vers f. On peut aussi encadrer f par des fonctions en escalier m_i (resp. M_i) minorantes (resp. majorantes) et dire que $I(f)$ existe si la borne supérieure des $I(m_i)$ est égale à la borne inférieure des $I(M_i)$.

On peut aussi introduire des subdivisions d de $[ab]$ ($d_o = a < d_1 < \ldots < d_{n+1} = b$) de pas $p(d) = \sup_i (d_{i+1} - d_i)$ et les sommes de Riemann

$$\Sigma(d) = \sum_{i=o}^{n} f(x_i)\ (d_{i+1} - d_i)$$

où $x_i \in [d_i , d_{i+1} [$

ainsi que les sommes de Darboux

$$s(d) = \sum_{i=o}^{n} m_i(d_{i+1} - d_i) \quad \text{où} \quad m_i = \inf_{x \in [d_i\, d_{i+1}]} f(x)$$

$$S(d) = \sum_{i=o}^{n} M_i(d_{i+1} - d_i) \quad \text{où} \quad M_i = \sup_{x \in [d_i\, d_{i+1}]} f(x)$$

On dit alors que f est intégrable sur $[ab]$ si la limite de $\Sigma(d)$ sur toutes les subdivisions de pas tendant vers 0 existe ou si les $s(d)$ et les $S(d)$ ont une borne commune.

Les deux points de vue sont bien sûr sensiblement équivalents, le second est cependant plus facile à utiliser sur machine que le second (on maîtrise mieux $[ab]$ que le graphe de f). Pour simplifier on va aussi se restreindre à des <u>subdivisans régulières</u> (en n segments égaux) et comme l'écran du microordinateur utilisé affiche horizontalement $256 = 2^8$ points on prendra pour n une puissance de deux.

a) <u>La méthode des rectangles</u>

C'est la plus naïve, on utilise les sommes :

$$I(n) = \sum_{i=o}^{n-1} f(a + \frac{i(b-a)}{n}) \cdot \frac{b-a}{n}$$

pour n une puissance de deux et on trace en plein les rectangles correspondants.

On affiche en bas à gauche de l'écran la valeur de l'approximation obtenue.

C'est une méthode rapide, mais assez fruste ne donnant aucune indication sur l'erreur commise (sauf que par exemple c'est une approximation par défaut si la fonction est croissante).

b) La méthode des sommes de Darboux

C'est là que l'on voit le mieux la convergence de l'intégrale quand on fait diminuer le pas.

Sur l'intervalle $[x_i \; x_{i+1}]$ on calcule le inf et le sup de $f(x)$ et on trace en plein le rectangle de sommets $(x_i, 0)$ $(x_{i+1}, 0)$, $(x_i, \inf)$ et $(x_{i+1}, \inf)$ et seulement les bords du rectangle de sommets $(x_i, 0)$, $(x_{i+1}, 0)$ $(x_i, \sup)$, $(x_{i+1}, \sup)$ et on affiche en bas à gauche

$$i = s(d) \qquad\qquad I = S(d)$$

pour la subdivision d.

Quand le pas $p(d)$ diminue, on voit

- d'une part sur l'écran les rectangles différences entre les "pleins" et les "vides" dont l'aire diminue.
- d'autre part l'écart entre les valeurs affichées de i et I se resserrer et suggérer la convergence de l'intégrale.

Même si le procédé n'est pas rapide (il faut calculer inf et sup) ni très précis (voir les exemples), il a le gros avantage de fournir un encadrement sûr de $I(f)$.

c) La méthode des sommes de Riemann

Là aussi on "aide" un peu le hasard en choisissant des subdivisions équiréparties, et en tirant (au sort avec le générateur aléatoire de la machine) un point x_i dans $[d_i\ d_{i+1}]$, et on affiche la valeur correspondante $I = \Sigma(d)$ en bas à gauche de l'écran et des rectangles sur le tracé de la courbe. C'est une méthode d'éxécution très rapide et l'intérêt est de la recommencer un certain nombre de fois avec le même pas pour voir la variation de $I(d)$ avec les tirages au sort, ce qui suggère une approche probabiliste de calcul.

2. - Approximation numérique d'une intégrale

Les méthodes précédentes permettent d'avoir une idée sur la façon dont convergent les sommes permettant de définir une intégrale mais sont assez mauvaises pour obtenir une approximation numérique de la dite intégrale si on sait qu'elle existe. C'est pourquoi j'ai ajouté deux méthodes numériquement plus efficaces et très classiques :

a) La méthode des trapèzes

La valeur numérique obtenue est affichée en bas à gauche et les trapèzes sont effictement tracés sur l'écran, ce qui permet de se rendre compte que :

- si la fonction f oscille beaucoup, ou si f a des discontinuités et si la subdivision est mal choisie la méthode des trapèzes laisse à désirer et fait un peu n'importe quoi.

- si au contraire f oscille peu et est continue, très rapidemement les trapèzes (ou du moins le côté supérieur) se confondent sur l'écran avec la courbe et on se convainc aisément que la méthode donne une bonne approximation.

b) La méthode de Simpson

Elle consiste, rappelons le, à approcher la courbe non par des segments de droite mais par des arcs de parabole déterminés par 3 points "consécutifs" de la courbe. L'approximation est meilleure que dans le cas des trapèzes mais les défauts, les cas de mauvais fonctionnement restent les mêmes : c'est pourquoi il n'y a pas d'illustration graphique.

3. - La réalisation pratique

La liste du programme est jointe en annexe à la fin. Le microordinateur utilisé est un modèle familial bas de gamme Sinclair Spectrum, muni d'une résolution graphique de 176 sur 256.

Le langage utilisé est une extension (Beta - Basic version 3.0) du Basic de base qui en conservant la facilité d'utilisation habituelle du Basic, présente l'énorme avantage de disposer de procédures avec passage de paramètres ce qui facilite grandement l'écriture d'un programme modulaire et augmente beaucoup la lisibilité du programme. L'existence de (pseudo) fenêtres permet aussi de mieux gérer l'écran et d'améliorer la lisibilité de l'affichage (par exemple en affichant en bas à droite les bornes : a, b, $m = \inf(f(x), x \in [a.b])$ et $M = \sup(f(x), x \in [ab])$).

Par exemple :

- les procédures début 1 et début 2 mettent en place les fenêtres et permettent l'introduction de la fonction (sous forme d'une chaine de caractères a$) et des bornes.

- calcul calcule les valeurs aux points $a + \frac{i(b-a)}{256}$ pour $i = 0, \ldots, 256$ et les met dans un tableau a( ).

- insup calcule le inf et le sup et est utilisé à de nombreuses reprises.

- dessin trace le graphe de f (compte tenu des options de cadrage).

- boite vide et boite pleine tracent les rectangles.

- menu procédure centrale permet via un choix offert à l'écran de faire exécuter la partie du programme voulue en appuyant sur une touche.

- options permet de choisir les options de cadrage (pour la lisibilité il est préférable que la fonction soit à valeurs positives et il peut être intéressant de translater f).

- enfin les procédures rectangles, darboux, riemann, trapèzes et simpson réalisent les processus décrits plus haut et se terminent par un retour au menu (après avoir demandé si on le désirait, une copie sur papier de l'écran).

Des commentaires sont prévus : ils sont stockés à part sur la microdisquette et chargés le cas échéant. Lors de la présentation en section expérimentale à Paris 6 ils n'ont pas été utilisés, les commentaires ayant été faits oralement au tableau par un enseignant.

4. - Les limites

Elles sont de plusieurs ordres. Il y en a qui sont théoriques et incontournables : la fonction se réduit pour l'ordinateur à un certain nombre de valeurs (comme c'est d'ailleurs le cas des fonctions rencontrées dans les sciences expérimentales) et il ne faut pas chercher à trop tirer de l'illustration proposée : on ne peut que suggérer une divergence ou une convergence : la fonction indicatrice de $\mathbb{Q}$ est hors de portée et la fonction $x \longmapsto \sin 256\, x$ sur $[0\ \pi]$ apparaîtra comme la fonction nulle.

D'autres sont liées au langage de programmation choisi et au matériel utilisé : un microordinateur plus puissant programmé en Pascal permettrait une réaction plus rapide (ce qui serait très agréable pour étudier l'évolution de l'intégrale en fonction des bornes, notamment au voisinage d'un point de discontinuité).

Il faut faire attention aussi à ne pas faire calculer à la machine des expressions dépourvues de sens (cf. l'exemple de $f(x) = x \sin 1/x$).

Enfin, et c'est un peu paradoxal, les fonctions en escalier sont difficiles à entrer (le programme n'accepte que les fonctions définies par $f(x)$ = expression en x) et mal traitées (à cause du pas égal à une puissance de 2 qui place mal la subdivision par rapport aux discontinuités éventuelles).

5. - Les exemples

Il s'agit de ceux montrés aux étudiants de la section expérimentale

a) $f(x) = \sqrt{1 + x^3}$ a été choisie comme exemple de fonction n'ayant pas de primitive s'exprimant avec les fonctions élémentaires (cf. la géométrie de la courbe $y^2 = x^3 + 1$).

b) $f(x) = x^2$ modulo 2 a été choisie pour ses discontinuités mettant en évidence les difficultés de convergence (surtout avec un pas constant) et parce que le calcul exact de $I(f)$ permettait de "vérifier" les résultats numériques

c) la sinusoïde amortie $f(x) = \sin 8x \exp(-x)$ permettait d'introduire des valeurs négatives et de se rendre compte de l'intérêt d'une translation

d) $f(x) = x \sin \frac{1}{x}$ était le type d'une fonction continue mais peu "régulière"

e) $f(x) = \frac{\cos 2x}{\sin x}$ sur $[-\pi, \pi]$ a été demandée par un étudiant. Les bornes choisies ont été - 3,15 et + 3,16 pour voir les asymptotes, à part celles-ci le graphe est très plat compte tenu de l'échelle (minimum des points tracés : - 61 ,.. et maximum : 199 ,...). C'est le seul exemple fourni d'intégrale divergente.

Quelques écrans

INTEGRATION

MENU

1 Approximation par des rect
angles
2 Sommes de Darboux
3 Sommes de Riemann
4 Methode des trapezes
5 Methode de Simpson
6 changement de bornes
7 Commentaires
8 Retour au choix des options
9 Nouvelle fonction
10 Fin

Options

On ne peut tracer les rectangl
es ou les trapezes que si l'axe
y=0 est sur l'ecran.Si inf>0 ou
si sup<0 on ne peut pas tracer
la courbe avec inf en bas de l'e
cran et sup en haut,de plus les
illustrations sont plus 'parlant
es' si l'origine est en bas de l
'ecran.

1 y=0 au milieu

2 y=0 en bas

3 y=0 en haut

4 y=0 en bas(on utilise f-inf)

5 y=0 en bas (on utilise f+con
st)

Les choix proposés par le menu sont sélectionnés en entrant au clavier le numéro retenu (et en validant par "ENTER", de façon à permettre une correction éventuelle).

Le retour au "menu" est automatique après chaque procédure.

Les "options" permettent un cadrage optimal (certains choix sont interdits par le programme) en faisant éventuellement une translation, et pour le tracé il faut que y = 0 soit sur l'écran.

$$\int_0^3 \sqrt{1+x^3}\,dx$$

Simpson
(pas : 16)



Fonction:
SQR (1+x*x*x)
I: 7,34139

Bornes:
a= 0.00 b= 3.00
m: 1.00 M: 5.29

La courbe est très régulière et la méthode de Simpson converge très vite (pour un pas de 2 on trouve : 7,34141)

Rectangles
Pas 32



Fonction:
SQR (1+x*x*x)
I= 6.567

Bornes:
a= 0.00 b= 3.00
m: 1.00 M: 5.29

La méthode assez primitive "des rectangles" est ici assez bonne. De plus la courbe étant monotone on sait que l'on a là, une estimation par défaut.

Rectangles
Pas 8



Fonction:
SQR (1+x*x*x)
I= 7.142

Bornes:
a= 0.00 b= 3.00
m: 1.00 M: 5.29

Pour $\int_0^3 \sqrt{1+x^3}\,dx$ tous les exemples sont réalisés sous l'option 2



$$\int_0^3 \sqrt{1+x^3}\,dx$$

Sommes de Darboux :

Quand le pas diminue on voit très nettement l'encadrement par les sommes minorantes et majorantes se resserrer.

pas 64



pas 32



pas 16



pas 8



pas 4



$$\int_0^3 \sqrt{1+x^3}\,dx$$

Sommes de Riemann
pas 32
5 tirages parmi d'autres.



Fonction: Bornes:
SQR (1+x*x*x) a= 0.00b= 3.00
i= 7.358 m: 1.00M: 5.29



Fonction: Bornes:
SQR (1+x*x*x) a= 0.00b= 3.00
i= 7.074 m: 1.00M: 5.29

Fonction: Bornes:
SQR (1+x*x*x) a= 0.00b= 3.00
i= 6.947 m: 1.00M: 5.29



Fonction: Bornes:
SQR (1+x*x*x) a= 0.00b= 3.00
i= 7.313 m: 1.00M: 5.29



Fonction: Bornes:
SQR (1+x*x*x) a= 0.00b= 3.00
i= 7.796 m: 1.00M: 5.29

$$\int_0^3 \sqrt{1+x^3}\,dx$$

Méthode des trapèzes

pas 64



pas 32



pas 4



Pour une telle fonction la méthode est excellente, donne rapidement une bonne approximation (par excès grâce à la convexité de la courbe) et on comprend qu'il soit vain d'espérer visualiser les arcs de parabole de la méthode de Simpson.

$$\int_0^3 (x^2 \bmod 2)\, dx$$

Simpson pas 16



Simpson avec un pas 2 (le plus petit possible) donne 2,375 ce qui n'est pas fameux

La fonction x^2 modulo 2 présente des discontinuités en $\sqrt{2}$, 2, $\sqrt{6}$, $\sqrt{8}$, ...

La valeur (exacte) de l'intégrale entre 0 et 3 se calcule très facilement, c'est

$$-11 + 6\sqrt{2} + 2\sqrt{6}$$
$$= 2,3842..$$

La méthode des trapèzes, si efficace dans l'exemple précédent est ici mal adaptée car le découpage ne tient pas compte des points de discontinuité.

(Les tracés de $\int_0^3 (x^2 \bmod 2)\, dx$ sont réalisés sous l'option 1).

Trapèzes pas 64



Trapèzes pas 16



Trapèzes pas: 32



Trapèzes pas 4



x^2 modulo 2 entre 0 et 3

Darboux pas 64



Les sommes de Darboux donnent au début des encadrements très larges (pour le pas 64 le 0,05 est dû à une erreur d'arrondi on devrait avoir 0) qui progressivement se resserrent quand le pas diminue

Au voisinage d'un point de discontinuité l'écart entre inf et sup reste de 2 mais quand le pas diminue, la largeur du rectangle diminue et l'erreur aussi : on passe de

$$0 < I < 4,88$$

à

$$2,20 < I < 2,58$$

Darboux pas 32



Darboux pas 16



Darboux pas 4



Darboux pas 8



Darboux pas 2



$$\int_0^3 (x^2 \bmod 2)\,dx$$

rectangles
pas 16



La fonction à intégrer n'étant pas régulière la méthode des rectangles ne donne a priori rien d'intéressant et est signalée pour mémoire.

On trouve en bas 4 exemples, pris au hasard, de la méthode de Riemann avec pas 16 : les résultats soutiennent aisément la comparaison avec ce que donnait Simpson.

Riemann
pas 16



Riemann
pas 16



Riemann
pas 16



Riemann
pas 16



$$\int_0^3 \sin 8x \cdot e^{-x}\, dx$$

(option 1)

La méthode de Simpson et celle des trapèzes convergent assez vite.

La méthode des rectangles est, comme avant, peu convaincante.

Simpson 4



Simpson 16



Rectangles pas 32



Trapèzes pas 16



Trapèzes pas 4



$$\int_0^3 e^{-x} \sin 8x \, dx$$

Sommes de Darboux (Option 1)

La fonction change de signe et l'illustration est moins parlante : dans les parties négatives la courbe disparaît et le "inf" cache le "sup". Dans les exemples du bas, l'utilisation de l'impression "OVER 1" améliore un peu les choses : pour imprimer un point on l'allume s'il est éteint et on l'éteint s'il est allumé. (Mise en marche par en donnant à la variable "un" la valeur 1).

pas 32



Fonction: Bornes:
SIN (8*x)/EXP x a= 0.00b= 3.00
i: -0.49I: 0.69m: -0.56M: 0.83

pas 32



Fonction: Bornes:
SIN (8*x)/EXP x a= 0.00b= 3.00
i: -0.49I: 0.69m: -0.56M: 0.83

pas 16



Fonction: Bornes:
SIN (8*x)/EXP x a= 0.00b= 3.00
i: -0.35I: 0.54m: -0.56M: 0.83

pas 8

Fonction: Bornes:
SIN (8*x)/EXP x a= 0.00b= 3.00
i: -0.11I: 0.34m: -0.56M: 0.83

pas 2



Fonction: Bornes:
SIN (8*x)/EXP x a= 0.00b= 3.00
i: .06I: 0.18m: -0.56M: 0.83

$$\int_0^3 e^{-x} \sin 8x \, dx$$

Sommes de Darboux (option 5)

pas 32



SIN (8*x)/EXP x+Bornes:
1 a= 0.00b= 3.00
i: 2.51I: 3.69m: 0.44M: 1.83

pas 8



SIN (8*x)/EXP x+Bornes:
1 a= 0.00b= 3.00
i: 2.89I: 3.34m: 0.44M: 1.83

pas 4



SIN (8*x)/EXP x+Bornes:
1 a= 0.00b= 3.00
i: 3.01I: 3.23m: 0.44M: 1.83

Pour améliorer la lisibilité en utilisant l'option 5 on remplace $e^{-x} \sin 8x$ par $e^{-x} \sin 8x + 1$ qui est positif sur $[0 \; 3]$. L'intégrale est augmentée de 3 et l'écart entre sommes majorantes et minorantes est inchangé.

L'affichage des valeurs est corrigé.

$\int_0^3 \sin 8x \cdot e^{-x}\,dx$

Sommes de Riemann :

pas 32

5 tirages au sort.

(option 1).



```
Fonction:          Bornes:
SIN (8*x)/EXP x    a=  0.00b=  3.00
i=   .035          m: -0.56M:  0.83
```



```
Fonction:          Bornes:
SIN (8*x)/EXP x    a=  0.00b=  3.00
i=  -.019          m: -0.56M:  0.83
```



```
Fonction:          Bornes:
SIN (8*x)/EXP x    a=  0.00b=  3.00
i= -0.122          m: -0.56M:  0.83
```



```
Fonction:          Bornes:
SIN (8*x)/EXP x    a=  0.00b=  3.00
i=  0.173          m: -0.56M:  0.83
```



```
Fonction:          Bornes:
SIN (8*x)/EXP x    a=  0.00b=  3.00
i=  0.290          m: -0.56M:  0.83
```

$$\int_{-0,2}^{0,2} x \sin \frac{1}{x}\, dx.$$

A priori la fonction est définie par 0 en 0 mais pour éviter des complications, on remplace +0,2 par +0,2001 et comme cela l'ordinateur ne calcule pas de valeur en 0 quand il remplit son tableau de 257 valeurs.



Simpson (pas : 4).



Méthode de Riemann (pas 16) :

2 exemples.



$$\int_{-0,2}^{0,2} x \sin 1/x \, dx$$

Sommes de Darboux



Fonction: Bornes:
x*SIN (1/x) a= -0.20b= 0.20
i: -.02I: .01m: -0.19M: 0.13

En option 1 l'illustration est assez peu parlante, sous l'option 5 une translation de 0,25 qui augmente l'intégrale de 0,1 permet de mieux voir la "convergence" de l'intégrale.



Fonction: Bornes:
x*SIN (1/x)+0.25a= -0.20b= 0.20
i: .08I: 0.12m: .06M: 0.38



Fonction: Bornes:
x*SIN (1/x)+0.25a= -0.20b= 0.20
i: .08I: 0.11m: .06M: 0.38



Fonction: Bornes:
x*SIN (1/x)+0.25a= -0.20b= 0.20
i: .09I: 0.11m: .06M: 0.38

$$\int_{-\pi}^{\pi} \frac{\cos 2x}{\sin x}\,dx$$

Simpson (pas 16)



(le pas 2 donne : 7,9...)

Les bornes choisies sont $-3,15$ et $+3,16$ pour éviter de planter le programme par un calcul de $\frac{1}{\sin x}$ en $0, \pm\pi$.

L'intégrale est divergente (mais compte tenu de la symétrie, on pourrait définir une "valeur principale de Cauchy" nulle).

Sommes de Darboux

Option 1

pas 16



pas 4



Option 5 (translation de 70)

pas 16



pas 4



- Le programme -

```
10 CLS 0
20 LET un=0
30 BORDER 1
40 titre
50 DIM a(257)
60 debut1
70 debut2
80 calcul
90 infsup 1,256,inf,sup
100 options a()
110 STOP
120
130
140 DEF PROC debut1
150 CLS 0
160 WINDOW 1,0,175,256,152
170 WINDOW 1: INK 2 : PAPER 6:
CLS
180 INPUT "entrer la fonction c
omme une variable de x",a$
190 DEF FN g(x)=VAL a$
200 WINDOW 2,0,23,128,16
210 WINDOW 4,0,7,128,8
220 WINDOW 2 : INK 0: PAPER 4:
CLS
230 WINDOW 4 : INK 0: PAPER 5:
CLS
240 WINDOW 3,128,23,128,16
250 WINDOW 5,128,7,128,8
260 WINDOW 3 : INK 0 : PAPER 5:
CLS
270 WINDOW 5 : INK 0 : PAPER 4:
CLS
280 PRINT WINDOW 2;"Fonction:"
290 PRINT WINDOW 2;"";a$
300 END PROC
310
320
330 DEF PROC debut2
340 INPUT "borne inferieure";a1
350 INPUT "borne superieure";b
360 CLS 3
370 CLS 5
380 WINDOW 3 : PRINT "Bornes:"
390 PRINT "a="; USING "###.##";
a1;"b="; USING "###.##";b
400 WINDOW 1,0,175,256,152
410 WINDOW 1: INK 2 : PAPER 6:
CLS
420 END PROC
430
440
450 DEF PROC calcul
460 LOCAL i
470 WINDOW 1
480 PRINT AT 9,7; CSIZE 16,24;"
PATIENCE!"
490 PRINT AT 15,9; CSIZE 8;"Cal
culs en cours"
500 FOR i=0 TO 256
510 LET a(i+1)=FN g(a1+i*(b-a1)
/256)
520 NEXT i
530 END PROC
540
550
560 DEF PROC infsup d,l, REF in
f, REF sup
570 LOCAL i
580 LET inf=a(d): LET sup=a(d)
590 FOR i=d+1 TO d+l
600 IF a(i)<inf THEN LET inf=a(
i)
610 IF a(i)>sup THEN LET sup=a(
i)
620 NEXT i
630 END PROC
640
650
660 DEF PROC dessin
670 CLS 1
680 WINDOW 1
690 PLOT 0,24+150*(a(1)-infn)/a
mp
700 FOR i=1 TO 256
710 LET w=24+150*(a(i)-infn)/am
p
720 DRAW TO i-1,w
730 NEXT i
740 PLOT 0,zero: DRAW 255,0
750 END PROC
770
780 DEF PROC boitevide x,y,haut
larg
790 PLOT x,y
800 DRAW larg,0: DRAW 0,haut
810 DRAW OVER un,-larg,0: DRAW
0,-haut
820 END PROC
830
840
850 DEF PROC boitepleine x,y,ha
ut,larg
860 LOCAL i
870 OVER un
880 FOR i=x TO x+larg
890 PLOT i,y
900 DRAW 0,haut
910 NEXT i
920 OVER 0
930 END PROC
940
950
960 DEF PROC rectangles
970 LOCAL i
980 LOCAL si
990 dessin
1000 WINDOW 1
1010 LET si=0
1020 INPUT "pas ? (4,8,16,32,64)
";p
1030 FOR i=1 TO 257-p STEP p
1040 boitepleine i-1,zero,150*a(
i)/amp,p-1
1050 LET si=si+a(i)
1060 NEXT i
1070 LET si=si*p*(b-a1)/256
1080 IF ABS si<0.001 THEN LET si
=0
1090 CLS 4
1100 PRINT WINDOW 4;"I="; USING
"###.###";si
1110 INPUT "retour au menu",q$
1120 IF q$="c" THEN COPY
1130 menu
1140 END PROC
1150
1160
1170 DEF PROC darboux
1180 LOCAL i
1190 WINDOW 1
1200 dessin
1210 INPUT "entrer le pas (4,8,1
6,32,64)";p
1220 LET si=0: LET sm=0
1230 LET p1=p
1240 FOR i=1 TO 257-p STEP p
1250 infsup i,p,min,max
1260 IF i+p=257 THEN LET p=p-1
1270 boitepleine i-1,zero,150*mi
n/amp,p
1280 boitevide i-1,zero,150*max/
amp,p
1290 LET si=si+min: LET sm=sm+ma
x
1300 NEXT i
1310 LET si=si*(b-a1)/256*p1: LE
T sm=(b-a1)*sm*p1/256
1320 IF ABS si<0.01 THEN LET si=
0: IF ABS sm<0.01 THEN LET sm=0
1330 CLS 4
1340 PRINT WINDOW 4;"i:"; USING
"###.##";si;"I:"; USING "###.##"
;sm
1350 INPUT "retour au menu";q$
1360 IF q$="c" THEN COPY
1370 menu
1380 END PROC
1390
```

```
1410
1420 DEF PROC trapezes
1430 LOCAL i
1440 LOCAL si
1450 CLS 1
1460 dessin
1470 LET si=0
1480 INPUT "pas? (4,8,16,32,64)"
;p
1490 LET p1=p
1500 FOR i=1 TO 257-p STEP p
1510 IF i+p=257 THEN LET p=p-1
1520 PLOT i-1,zero: DRAW 0,150*a
(i)/amp: DRAW p,150*(a(i+p-1)-a(
i))/amp: DRAW TO i+p-1,zero: DRA
W TO i-1,zero
1530 LET si=si+a(i)
1540 NEXT i
1550 LET si=si+(a(257)-a(1))/2
1560 LET si=si*p1*(b-a1)/256
1570 IF ABS si<10E-4 THEN LET si
=0
1580 CLS 4: PRINT WINDOW 4; "I:"
; USING "###.####";si
1590 INPUT "retour au menu";q$
1600 IF q$="c" THEN COPY
1610 menu
1620 END PROC
1630
1640
1650 DEF PROC riemann
1660 LOCAL si
1670 LOCAL j
1680 LOCAL i
1690 WINDOW 1: CLS
1700 dessin
1710 INPUT "pas? (4,8,16,32,64)"
;p
1720 LET si=0
1730 LET p1=p
1740 FOR i=1 TO 257-p STEP p
1750 LET j=i+ RNDM(p-1)
1760 IF i+p=257 THEN LET p=p-1
1770 boitevide i-1,zero,150*a(j)
/amp,p
1780 LET si=si+a(j)
1790 NEXT i
1800 IF ABS si<0.001 THEN LET si
=0
1810 LET si=si*p1*(b-a1)/256
1820 CLS 4: PRINT WINDOW 4;"i=";
 USING "###.###";si
1830 INPUT "retour au menu";q$
1840 IF q$="c" THEN COPY
1850 menu
1860 END PROC
1870
1880 DEF PROC commentaires
1890 MERGE "com"
1900 GO SUB 3280
1910 DELETE 3280 TO 3280
1920 menu
1930 END PROC
1940
1950 DEF PROC simpson REF a()
1960 WINDOW 1: CLS
1970 dessin
1980 INPUT "pas? (2,4,8,16,32,64
)";p
1990 LET si=a(1)+a(257)
2000 FOR i=p+1 TO 257-p STEP 2*p
2010 LET si=si+4*a(i)
2020 NEXT i
2030 FOR i=2*p+1 TO 257-2*p STEP
 2*p
2040 LET si=si+2*a(i)
2050 NEXT i
2060 LET si=si*p*(b-a1)/768
2070 IF ABS si < 10E -5 THEN LET
 si=0
2080 CLS 4: PRINT WINDOW 4;"  I:
"; USING "###.#####";si
2090 INPUT "retour au menu";q$
2100 IF q$="c" THEN COPY
2110  menu
2120 END PROC
2130
2140 DEF PROC titre
2150 FOR i=4 TO 20 STEP 4
2160 CLS
2170 PRINT AT 9,12-i/2; CSIZE i,
i;"INTEGRATION"
2180 PAUSE 30
2190 NEXT i
2200 PAUSE 40
2210 FOR i=1 TO 240
2220 SCROLL 9
2230 NEXT i
2240 CLS
2250 PRINT AT 5,4;"Ce programme
cherche a illustrer certains asp
ects de la theorie de l'integrat
ion : approximation par des rect
angles ou des  trapezes, evoluti
on de l'approximation avec la lo
ngueur  du pas d'integration ou
lechoix des bornes , calcul nume
rique d'integrales definies"
2260 INPUT "suite";q$
2270 IF q$="c" THEN COPY
2280 END PROC
2290
2300
2310 DEF PROC menu
2320 CLS 1
2330 LOCAL i
2340 WINDOW 1
2350 PRINT AT 3,9; CSIZE 16,16;"
MENU"
2360 PRINT AT 7,2;"1  Approximat
ion par des rect angles"
2370 PRINT AT 9,2;"2  Sommes de
Darboux"
2380 PRINT AT 10,2;"3  Sommes de
 Riemann"
2390 PRINT AT 11,2;"4  Methode d
es trapezes "
2400 PRINT AT 12,2;"5  Methode d
e Simpson"
2410 PRINT AT 13,2;"6  changemen
t de bornes"
2420 PRINT AT 14,2;"7  Commentai
res"
2430 PRINT AT 15,2;"8  Retour au
 choix des options"
2440 PRINT AT 16,2;"9  Nouvelle
fonction"
2450 PRINT AT 17,1;"10  Fin"
2460 INPUT i
2470 IF i=1 THEN  rectangles
2480 IF i=2 THEN  darboux
2490 IF i=3 THEN  riemann
2500 IF i=4 THEN  trapezes
2510 IF i=5 THEN  simpson a()
2520 IF i=6 THEN GO TO 70
2530 IF i=7 THEN  commentaires
2540 IF i=8 THEN  options a()
2550 IF i=9 THEN GO TO 80
2560 IF i=10 THEN STOP
2570 END PROC
2580
2590
```

```
2600
2610 DEF PROC options REF a()
2620 CLS 5
2630 LET ifn=inf
2640 LET spu=sup
2650 IF ABS (ifn)<0.01 THEN LET
ifn=0
2660 IF ABS (spu)<0.01 THEN LET
spu=0
2670 WINDOW 5: PRINT "m:"; USING
 "###.##";ifn;"M:"; USING "###.#
#";spu
2680 WINDOW 1 : CLS
2690 PRINT AT 3,8; CSIZE 16,16;"
Options"
2700 PRINT AT 8,2;"On ne peut tr
acer les rectangles ou les trape
zes que si l'axe y=0 est sur l'e
cran.Si inf>0 ou si sup<0 on ne
peut pas tracer  la courbe avec
inf en bas de l'ecran et sup en
haut,de plus les illustrations s
ont plus 'parlantes' si l'origin
e est en bas de l'ecran."
2710 INPUT "suite";q$
2720 CLS 1
2730 PRINT AT 3,2;"1 y=0 au mili
eu"
2740 PRINT AT 6,2;"2 y=0 en bas"
2750 PRINT AT 9,2;"3 y=0 en haut
"
2760 PRINT AT 12,2;"4 y=0 en bas
(on utilise f-inf)"
2765 PRINT AT 15,2;"5 y=0 en bas
 (on utilise f+const)"
2770 INPUT i
2780 CLS 1
2790 IF i=1 THEN  option1 a()
2800 IF i=2 THEN  option2 a()
2810 IF i=3 THEN  option3 a()
2820 IF i=4 THEN  option4 4, a()
2825 IF i=5 THEN  option4 5, a()
2830 END PROC
2840
2850
2860 DEF PROC option1 REF a()
2870 IF inf>0 OR sup<0 THEN PRIN
T AT 9,0; CSIZE 16,16;"Option in
terdite": PAUSE 80:  options a()
2880 LET infn=inf: LET supn=sup
2890 LET amp=sup-inf
2900 LET zero=24-150*infn/amp
2910 menu
2920 END PROC
2930
2940
2950 DEF PROC option2 REF a()
2960 IF inf<0 THEN PRINT AT 9,0;
 CSIZE 16,16;"Option interdite":
 PAUSE 80: options a()
2970 LET infn=0: LET supn=sup
2980 LET amp=supn
2990 LET zero=24
3000 menu
3010 END PROC
3020
3030
3040 DEF PROC option3 REF a()
3050 IF sup>0 THEN PRINT AT 9,0;
 CSIZE 16,16;"Option interdite":
 PAUSE 80: options a()
3060 LET infn=inf: LET supn=0
3070 LET amp=-infn
3080 LET zero=174
3090 menu
3100 END PROC
3110
3120
3130 DEF PROC option4 c, REF a()
3140 LOCAL i
3142 IF c=4 THEN LET t=inf : ELS
E INPUT "valeur de la translatio
n(signe celui de m):";t
3150 FOR i=1 TO 257
3160 LET a(i)=a(i)-t
3170 NEXT i
3180 LET infn=0: LET supn=sup-t
3190 LET amp=supn
3200 LET zero=24
3210 LET sup=supn: LET inf=inf-t
3220 IF t>0 THEN LET a$=a$+"-":
ELSE LET a$=a$+"+"
3221 CLS 5
3222 LET ifn=inf
3223 LET spu=sup
3224 IF ABS (ifn)<0.01 THEN LET
ifn=0
3225 IF ABS (spu)<0.01 THEN LET
spu=0
3226 WINDOW 5: PRINT "m:"; USING
 "###.##";ifn;"M:"; USING "###.#
#";spu
3230 CLS 2
3240 PRINT WINDOW 2;"Fonction:"
3250 PRINT WINDOW 2;a$;ABS t
3260 menu
3270 END PROC
```